Pensieve header: This is a pruned version of SL2PortfolioProgram.nb and SL2PortfolioTesting.nb from http://drorbn.net/ap/Projects/SL2Portfolio/.

SL2Portfolio

Initialization / Utilities

SL2Portfolio

In[∘]:=
```
$p = 2; $k = 1; $U = QU; $E := {$k, $p};
$trim := {ℏ^(p_·) /; p > $p → 0, ϵ^(k_·) /; k > $k → 0};
qℏ = e^(γ ϵ ℏ);
T2t = {T_i^(p_·) → e^(p ℏ t_i), T^(p_·) → e^(p ℏ t)};
t2T = {e^(c_· t_i + b_·) :→ T_i^(c/ℏ) e^b, e^(c_· t + b_·) :→ T^(c/ℏ) e^b, e^(ℰ_) :→ e^(Expand@ℰ)};
SetAttributes[SS, HoldAll];
SS[ℰ_, op_] := Collect[
   Normal@Series[If[$p > 0, ℰ, ℰ /. T2t], {ℏ, 0, $p}],
   ℏ, op];
SS[ℰ_] := SS[ℰ, Together];
Simp[ℰ_, op_] := Collect[ℰ, _CU | _QU, op];
Simp[ℰ_] := Simp[ℰ, SS[#, Expand] &];
Kδ /: Kδ_(i_, j_) := If[i === j, 1, 0];
c_Integer_(k_Integer) := c + O[ϵ]^(k+1);
```

SL2Portfolio

In[∘]:=
```
CF[ℰ_] := ExpandDenominator@
   ExpandNumerator@Together[Expand[ℰ] //. e^(x_) e^(y_) :→ e^(x+y) /. e^(x_) :→ e^(CF[x])];
```

SL2Portfolio

In[∘]:=
```
Unprotect[SeriesData];
SeriesData /: CF[sd_SeriesData] := MapAt[CF, sd, 3];
SeriesData /: Expand[sd_SeriesData] := MapAt[Expand, sd, 3];
SeriesData /: Simplify[sd_SeriesData] := MapAt[Simplify, sd, 3];
SeriesData /: Together[sd_SeriesData] := MapAt[Together, sd, 3];
SeriesData /: Collect[sd_SeriesData, specs__] := MapAt[Collect[#, specs] &, sd, 3];
Protect[SeriesData];
```

SL2Portfolio

DeclareAlgebra

SL2Portfolio

In[∘]:=
```
Unprotect[NonCommutativeMultiply]; Attributes[NonCommutativeMultiply] = {};
(NCM = NonCommutativeMultiply)[x_] := x;
NCM[x_, y_, z__] := (x ** y) ** z;
0 ** _ = _ ** 0 = 0;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
B[x_, y_, e_] := B[x, y, e] = B[x, y];
```

SL2Portfolio

In[∘]:=

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, g, cp, M, CE, k = 0,
    gs = Generators /. {opts},
    cs = Centrals /. {opts} /. Centrals → {} },
  (#_U = U@#) & /@ gs;
  gp = Alternatives @@ gs; gp = gp | gp_; (* gens *)
  sr = Flatten@Table[{g → ++k, g_{i_} → {i, k}}, {g, gs}]; (* sorting → *)
  cp = Alternatives @@ cs; (* cents *)
  SetAttributes[M, HoldRest]; M[0, _] = 0; M[a_, x_] := a x;
  CE[ℰ_] := Collect[ℰ, _U, Expand] /. $trim;
  U_{i_} [ℰ_] := ℰ /. {t : cp :> t_i, u_U :> (#_i &) /@ u};
  U_{i_} [NCM[]] = U@{} = 1_U = U[];
  B[U@(x_)_{i_}, U@(y_)_{i_}] := U_i@B[U@x, U@y];
  B[U@(x_)_{i_}, U@(y_)_{j_}] /; i =!= j := 0;
  B[U@y_, U@x_] := CE[-B[U@x, U@y]];
  x_ ** (c_. 1_U) := CE[c x]; (c_. 1_U) ** x_ := CE[c x];
  (a_. U[xx___, x_]) ** (b_. U[y_, yy___]) := If[OrderedQ[{x, y} /. sr],
    CE@M[a b /. $trim, U[xx, x, y, yy]],
    U@xx ** CE@M[a b /. $trim, U@y ** U@x + B[U@x, U@y, $E]] ** U@yy ];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := CE[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := CE[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := CE[c U@{r}];
  U@{l_Plus, r___} := CE[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  U[ℰ_NonCommutativeMultiply] := U /@ ℰ;
  𝕆_U[specs___, poly_] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List :> l_null, {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. l_{s_} :> (l /. x_{i_} :> x_s));
    CE[Total[
        CoefficientRules[poly, vs] /. (p_ → c_) :> c U@(us^p)
      ]] /. x_{null} :> x];
  𝕆_U[specs___, 𝔼[L_, Q_, P_]] := 𝕆_U[specs, SS@Normal[P e^(L+Q)]];
  σ_{rs___} [c_. * u_U] := (c /. (t : cp)_{j_} :> t_{j /. {rs}}) U[List @@ (u /. v_{j_} :> v_{j /. {rs}})];
  m_{j_→k_} [c_. * u_U] := CE[((c /. (t : cp)_j → t_k) DeleteCases[u, _{j|k}]) **
      U @@ Cases[u, w_j :> w_k] ** U @@ Cases[u, _k]];
  U /: c_. * u_U * v_U := CE[c u ** v];
  S_{i_} [c_. * u_U] := CE[((c /. S_i[U, Centrals]) DeleteCases[u, _i]) **
      U_i[NCM @@ Reverse@Cases[u, x_i :> S@U@x]]];
  Δ_{i_→j_,k_} [c_. * u_U] := CE[((c /. Δ_{i→j,k}[U, Centrals]) DeleteCases[u, _i]) **
      (NCM @@ Cases[u, x_i :> σ_{1→j,2→k}@Δ@U@x] /. NCM[] → U[])]; ]
```

SL2Portfolio

DeclareMorphism

SL2Portfolio

In[∘]:=
```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs,
   {(g_ → img_) ⧴ (m[U[g]] = img), (g_ ⧴ img_) ⧴ (m[U[g]] := img /. $trim)}, {1}];
  m[1_U] = 1_V;
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NCM @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /.  u_U ⧴ m[u]] /. $trim; )
```

SL2Portfolio

Meta-Operations

SL2Portfolio

In[∘]:=
```
σ_rs___[ℰ_Plus] := σ_rs /@ ℰ;
m_j_→j_ = Identity; m_j_→k_[0] = 0;
m_j_→k_[ℰ_Plus] := Simp[m_j→k /@ ℰ];
m_is___,i_,j_→k_[ℰ_] := m_j→k @ m_is,i→j @ ℰ;
S_i_[ℰ_Plus] := Simp[S_i /@ ℰ];
Δ_is___[ℰ_Plus] := Simp[Δ_is /@ ℰ];
```

SL2Portfolio

Implementing CU = $\mathcal{U}(sl_2^{\gamma\epsilon})$

SL2Portfolio

In[∘]:=
```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[a_CU, y_CU] = -γ y_CU; B[x_CU, a_CU] = -γ x_CU;
B[x_CU, y_CU] = 2 ϵ a_CU - t 1_CU;
(S@y_CU = -y_CU; S@a_CU = -a_CU; S@x_CU = -x_CU;)
S_i_[CU, Centrals] = {t_i → -t_i};
Δ@y_CU = CU@y_1 + CU@y_2; Δ@a_CU = CU@a_1 + CU@a_2; Δ@x_CU = CU@x_1 + CU@x_2;
Δ_i_→j_,k_[CU, Centrals] = {t_i → t_j + t_k};
```

SL2Portfolio

Implementing QU = $\mathcal{U}_q(sl_2^{\gamma\epsilon})$

SL2Portfolio

In[∘]:=
```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
B[a_QU, y_QU] = -γ y_QU; B[x_QU, a_QU] = -γ QU@x;
B[x_QU, y_QU] := SS[q_ħ - 1] QU@{y, x} + 𝕆_QU[{a}, SS[(1 - T ⅇ^(-2 ϵ a ħ)) / ħ]];
(S@y_QU := 𝕆_QU[{a, y}, SS[-T^-1 ⅇ^(ħ ϵ a) y]]; S@a_QU = -a_QU; S@x_QU := 𝕆_QU[{a, x}, SS[-ⅇ^(ħ ϵ a) x]];)
S_i_[QU, Centrals] = {t_i → -t_i, T_i → T_i^-1};
Δ@y_QU := 𝕆_QU[{y_1, a_1}_1, {y_2}_2, SS[y_1 + T_1 ⅇ^(-ħ ϵ a_1) y_2]];
Δ@a_QU = QU@a_1 + QU@a_2; Δ@x_QU := 𝕆_QU[{a_1, x_1}_1, {x_2}_2, SS[x_1 + ⅇ^(-ħ ϵ a_1) x_2]];
Δ_i_→j_,k_[QU, Centrals] = {t_i → t_j + t_k, T_i → T_j T_k};
```

SL2Portfolio

The representation ρ

SL2Portfolio

In[ ]:=
```
ρ@y_CU = ρ@y_QU = (0 0; ϵ 0); ρ@a_CU = ρ@a_QU = (γ 0; 0 0);
ρ@x_CU = (0 γ; 0 0); ρ@x_QU = (0  (1 - e^(-γ ϵ ℏ)) / (ϵ ℏ); 0  0);
ρ[e^𝓔_] := MatrixExp[ρ[𝓔]];
ρ[𝓔_] := (𝓔 /. T2t /. t → γ ϵ /. (U : CU | QU) [u___] :→ Fold[Dot, (1 0; 0 1), ρ /@ U /@ {u}])
```

SL2Portfolio

tSW

SL2Portfolio

Goal. In either U, compute $F = e^{-\eta y}\, e^{\xi x}\, e^{\eta y}\, e^{-\xi x}$. First compute $G = e^{\xi x}\, y\, e^{-\xi x}$, a finite sum. Now F satisfies the ODE $\partial_\eta F = \partial_\eta\left(e^{-\eta y}\, e^{\eta G}\right) = -yF + FG$ with initial conditions $F(\eta = 0) = 1$. So we set it up and solve:

SL2Portfolio

In[ ]:=
```
SW_xy[U_, kk_] :=
  SW_xy[U, kk] = Block[{$U = U, $k = kk, $p = kk}, Module[{G, F, fs, f, bs, e, b, es},
     G = Simp[Table[ξ^k / k!, {k, 0, $k + 1}].NestList[Simp[B[x_U, #]] &, y_U, $k + 1]];
     fs = Flatten@Table[f_(l,i,j,k)[η], {l, 0, $k}, {i, 0, l}, {j, 0, l}, {k, 0, l}];
     F = fs.(bs = fs /. f_(l_,i_,j_,k_)[η] :→ ϵ^l U@{y^i, a^j, x^k});
     es = Flatten[
       Table[Coefficient[e, b] == 0, {e, {F - 1_U /. η → 0, F ** G - y_U ** F - ∂_η F}}, {b, bs}]];
     F = F /. DSolve[es, fs, η][[1]];
     𝔼[0,
       ξ x + η y + (U /. {CU → -t η ξ, QU → η ξ (1 - T) / ℏ}),
       F + 0_$k /. {e- → 1, U → Times}
      ] /. (v : η | ξ | t | T | y | a | x) → v_1
    ]];
tSW_(xy_,i_,j_→k_) := SW_xy[$U, $k] /. {ξ_1 → ξ_i, η_1 → η_j, (v : t | T | y | a | x)_1 → v_k};
tSW_(xa,i_,j_→k_) := 𝔼[α_j a_k, e^(-γ α_j) ξ_i x_k, 1];
tSW_(ay,i_,j_→k_) := 𝔼[α_i a_k, e^(-γ α_i) η_j y_k, 1];
```

SL2Portfolio

Exponentials as needed.

SL2Portfolio

Task. Define $\mathrm{Exp}_{U_i,k}[\xi, P]$ which computes $e^{\xi\,\mathbb{O}(P)}$ to ϵ^k in the algebra U_i, where ξ is a scalar, X is x_i or y_i, and P is an ϵ-dependent near-docile element, giving the answer in $\mathbb{E}$-form. Should satisfy $U@\mathrm{Exp}_{U_i,k}[\xi, P] == \mathbb{S}_U\left[e^{\xi x},\ x \to \mathbb{O}(P)\right]$.

Methodology. If $P_0 := P_{\epsilon=0}$ and $e^{\xi\,\mathbb{O}(P)} = \mathbb{O}\left(e^{\xi P_0} F(\xi)\right)$, then $F(\xi = 0) = 1$ and we have:
$\mathbb{O}\left(e^{\xi P_0}(P_0\, F(\xi) + \partial_\xi F\right) = \mathbb{O}\left(\partial_\xi e^{\xi P_0} F(\xi)\right) = \partial_\xi\, \mathbb{O}\left(e^{\xi P_0} F(\xi)\right) = \partial_\xi e^{\xi\,\mathbb{O}(P)} = e^{\xi\,\mathbb{O}(P)}\, \mathbb{O}(P) = \mathbb{O}\left(e^{\xi P_0} F(\xi)\right) \mathbb{O}(P)$.
This is an ODE for F. Setting inductively $F_k = F_{k-1} + \epsilon^k \varphi$ we find that $F_0 = 1$ and solve for φ.

SL2Portfolio

In[∘]:=
```
(* Bug: The first line is valid only if 𝕆(e^P0) == e^𝕆(P0). *)
(* Bug: ξ must be a symbol. *)
Exp_{U_i_,0}[ξ_, P_] := Module[{LQ = Normal@P /. ϵ → 0},
   𝔼[ξ LQ /. (x | y)_i → 0, ξ LQ /. (t | a)_i → 0, 1]];
Exp_{U_i_,k_}[ξ_, P_] := Block[{$U = U, $k = k},
  Module[{P0, φ, φs, F, j, rhs, at0, atξ},
   P0 = Normal@P /. ϵ → 0;
   φs =
    Flatten@Table[φ_{j1,j2,j3}[ξ], {j2, 0, k}, {j1, 0, 2 k + 1 - j2}, {j3, 0, 2 k + 1 - j2 - j1}];
   F = Normal@Last@Exp_{U_i,k-1}[ξ, P] + ϵ^k φs.(φs /. φ_{js__}[ξ] :→ Times @@ {y_i, a_i, x_i}^{js});
   rhs = Normal@
     Last@m_{i,j→i}[𝔼[ξ P0 /. (x | y)_i → 0, ξ P0 /. (t | a)_i → 0, F + 0_k] m_{i→j}@𝔼[0, 0, P + 0_k]];
   at0 = (# == 0) & /@ Flatten@CoefficientList[F - 1 /. ξ → 0, {y_i, a_i, x_i}];
   atξ = (# == 0) & /@ Flatten@CoefficientList[(∂_ξ F) + P0 F - rhs, {y_i, a_i, x_i}];
   𝔼[ξ P0 /. (x | y)_i → 0, ξ P0 /. (t | a)_i → 0, F + 0_k] /.
    DSolve[And @@ (at0 ⋃ atξ), φs, ξ][[1]] ]]
```

SL2Portfolio

Zip and Bind

SL2Portfolio

In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
```

SL2Portfolio

In[∘]:=
```
{t*, y*, a*, x*, z*} = {τ, η, α, ξ, ζ};
{τ*, η*, α*, ξ*, ζ*} = {t, y, a, x, z};  (u_i_)* := (u*)_i;
```

SL2Portfolio

In[∘]:=
```
Zip_{}[P_] := P; Zip_{ζ_,ζs___}[P_] := (Expand[P // Zip_{ζs}] /. f_. ζ^d_. :→ ∂_{ζ*,d} f) /. ζ* → 0
```

SL2Portfolio

QZip implements the “Q-level zips” on $\mathbb{E}(L, Q, P) = P e^{L+Q}$. Such zips regard the L variables as scalars.

SL2Portfolio

In[∘]:=
```
QZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L, Q2, Det[qt] e^-Q2 Zip_{ζs}[e^Q1 (P /. zrule)]] ];
QZip_{ζs_List} := QZip_{ζs,CF};
```

SL2Portfolio

LZip implements the “L-level zips” on $\mathbb{E}(L, Q, P) = P e^{L+Q}$. Such zips regard all of $P e^{Q}$ as a single”P”. Here the z’s are t and α and the ζ’s are τ and a.

SL2Portfolio

In[ ]:=
```
LZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
   lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} L, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → lt.(zs + ys)];
   L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   Q2 = (Q1 = Q /. T2t /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L2, Q2, Det[lt] e^(-L2-Q2) Zip_ζs[e^(L1+Q1) (P /. T2t /. zrule)]] //. t2T ];
LZip_{ζs_List} := LZip_{ζs,CF};
```

SL2Portfolio

In[ ]:=
```
Bind_{}[L_, R_] := L R;
Bind_{is__}[L_𝔼, R_𝔼] := Module[{n},
   Times[
     L /. Table[(v : T | t | a | x | y)_i → v_{n@i}, {i, {is}}],
     R /. Table[(v : τ | α | ξ | η)_i → v_{n@i}, {i, {is}}]
   ] // LZip_{Flatten@Table[{τ_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Flatten@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]} ];
B_{L_List} := Bind_L; B_{is___} := Bind_{is};
Bind[ℰ_𝔼] := ℰ;
Bind[Ls__, ζs_List, R_] := Bind_ζs[Bind[Ls], R];
```

SL2Portfolio

Tensorial Representations

SL2Portfolio

In[ ]:=
```
tη = t𝟙 = 𝔼[0, 0, 1 + 0_{$k}];
```

SL2Portfolio

In[ ]:=
```
tm_{i_,j_→k_} := Module[{tk},
   𝔼[(τ_i + τ_j) t_k + α_i a_k + α_j a_k, η_i y_k + ξ_j x_k, 1]
    (tSW_{xy,i,j→tk} /. {t_tk → t_k, T_tk → T_k, y_tk → e^(-γ α_i) y_k, a_tk → a_k, x_tk → e^(-γ α_j) x_k})];
m_{j_→k_}[ℰ_𝔼] := ℰ ~ B_{j,k} ~ tm_{j,k→k};
```

SL2Portfolio

In[ ]:= $tm_{1,2\to 3}$

SL2Portfolio

Out[ ]=
$$\mathbb{E}\Big[a_3\,\alpha_1 + a_3\,\alpha_2 + t_3\,(\tau_1+\tau_2),\; y_3\,\eta_1 + e^{-\gamma\alpha_1}\,y_3\,\eta_2 + e^{-\gamma\alpha_2}\,x_3\,\xi_1 + \frac{(1-T_3)\,\eta_2\,\xi_1}{\hbar} + x_3\,\xi_2,$$
$$1 + \frac{1}{4\hbar}\,\eta_2\,\xi_1\,\big(8\,\hbar\,a_3\,T_3 + 4\,e^{-\gamma\alpha_1-\gamma\alpha_2}\,\gamma\,\hbar^2\,x_3\,y_3 + 2\,e^{-\gamma\alpha_1}\,\gamma\,\hbar\,y_3\,\eta_2 - 6\,e^{-\gamma\alpha_1}\,\gamma\,\hbar\,T_3\,y_3\,\eta_2 +$$
$$2\,e^{-\gamma\alpha_2}\,\gamma\,\hbar\,x_3\,\xi_1 - 6\,e^{-\gamma\alpha_2}\,\gamma\,\hbar\,T_3\,x_3\,\xi_1 + \gamma\,\eta_2\,\xi_1 - 4\,\gamma\,T_3\,\eta_2\,\xi_1 + 3\,\gamma\,T_3^2\,\eta_2\,\xi_1\big)\,\epsilon + O[\epsilon]^2\Big]$$

SL2Portfolio

In[∘]:=
```
S[U_, kk_] := S[U, kk] = Module[{OE},
    OE = m_{3,2,1→1}[Exp_{QU_1,$k}[η, S_1[QU[y_1]] /. QU → Times]
        Exp_{QU_2,$k}[α, S_2[QU[a_2]] /. QU → Times] Exp_{QU_3,$k}[ξ, S_3[QU[x_3]] /. QU → Times]];
    𝔼[-t_1 τ_1 + OE⟦1⟧, OE⟦2⟧, OE⟦3⟧] /. {η → η_1, α → α_1, ξ → ξ_1}];
tS_{i_} := S[$U, $k] /. {(v : τ | η | α | ξ)_1 → v_i, (v : t | T | y | a | x)_1 → v_i};
```

SL2Portfolio

In[∘]:= tS_1

SL2Portfolio

Out[∘]=
$$\mathbb{E}\Big[-a_1\alpha_1 - t_1\tau_1,\ \frac{-e^{\gamma\alpha_1}\hbar y_1\eta_1 - e^{\gamma\alpha_1}\hbar T_1 x_1\xi_1 + e^{\gamma\alpha_1}\eta_1\xi_1 - e^{\gamma\alpha_1}T_1\eta_1\xi_1}{\hbar T_1},$$
$$1 + \frac{1}{4\hbar T_1^2}\big(4e^{\gamma\alpha_1}\gamma\hbar^2 T_1 y_1\eta_1 - 4e^{\gamma\alpha_1}\hbar^2 a_1 T_1 y_1\eta_1 - 2e^{2\gamma\alpha_1}\gamma\hbar^2 y_1^2\eta_1^2 - 4e^{\gamma\alpha_1}\hbar^2 a_1 T_1^2 x_1\xi_1 -$$
$$4e^{\gamma\alpha_1}\gamma\hbar T_1\eta_1\xi_1 + 8e^{\gamma\alpha_1}\hbar a_1 T_1\eta_1\xi_1 + 4e^{\gamma\alpha_1}\gamma\hbar T_1^2\eta_1\xi_1 - 4e^{2\gamma\alpha_1}\gamma\hbar^2 T_1 x_1 y_1\eta_1\xi_1 +$$
$$6e^{2\gamma\alpha_1}\gamma\hbar y_1\eta_1^2\xi_1 - 2e^{2\gamma\alpha_1}\gamma\hbar T_1 y_1\eta_1^2\xi_1 - 2e^{2\gamma\alpha_1}\gamma\hbar^2 T_1^2 x_1^2\xi_1^2 + 6e^{2\gamma\alpha_1}\gamma\hbar T_1 x_1\eta_1\xi_1^2 -$$
$$2e^{2\gamma\alpha_1}\gamma\hbar T_1^2 x_1\eta_1\xi_1^2 - 3e^{2\gamma\alpha_1}\gamma\eta_1^2\xi_1^2 + 4e^{2\gamma\alpha_1}\gamma T_1\eta_1^2\xi_1^2 - e^{2\gamma\alpha_1}\gamma T_1^2\eta_1^2\xi_1^2\big)\,\epsilon + O[\epsilon]^2\Big]$$

SL2Portfolio

In[∘]:=
```
Δ[U_, kk_] := Δ[U, kk] = Module[{OE},
    OE = Block[{$k = kk, $p = kk + 1},
      m_{1,3,5→1}@m_{2,4,6→2}@Times[ (* Warning: wrong unless $p≥$k+1! *)
          ReplacePart[1 → 0]@Exp_{QU_1,$k}[η, Δ_{1→1,2}[QU[y_1]] /. QU → Times],
          ReplacePart[2 → 0]@Exp_{QU_3,$k}[α, Δ_{3→3,4}[QU[a_3]] /. QU → Times],
          ReplacePart[1 → 0]@Exp_{QU_5,$k}[ξ, Δ_{5→5,6}[QU[x_5]] /. QU → Times]
        ] /. {η → η_1, α → α_1, ξ → ξ_1}];
    𝔼[τ_1 (t_1 + t_2) + α_1 (a_1 + a_2), OE⟦2⟧, OE⟦3⟧]];
tΔ_{i_→j_,k_} :=
  Δ[$U, $k] /. {(v : τ | η | α | ξ)_1 → v_i, (v : t | T | y | a | x)_1 → v_j, (v : t | T | y | a | x)_2 → v_k};
```

SL2Portfolio

In[∘]:= $t\Delta_{1\to1,2}$

SL2Portfolio

Out[∘]=
$$\mathbb{E}\Big[(a_1+a_2)\alpha_1 + (t_1+t_2)\tau_1,\ y_1\eta_1 + T_1 y_2\eta_1 + x_1\xi_1 + x_2\xi_1,$$
$$1 + \frac{1}{2}\big(-2\hbar a_1 T_1 y_2\eta_1 + \gamma\hbar T_1 y_1 y_2\eta_1^2 - 2\hbar a_1 x_2\xi_1 + \gamma\hbar x_1 x_2\xi_1^2\big)\,\epsilon + O[\epsilon]^2\Big]$$

SL2Portfolio

The Faddeev-Quesne formula:

SL2Portfolio

In[∘]:=
$$e_{q\_,k\_}[x\_] := e\text{^}\left(\sum_{j=1}^{k+1}\frac{(1-q)^j x^j}{j\,(1-q^j)}\right);\quad e_{q\_}[x\_] := e_{q,\$k}[x]$$

SL2Portfolio

In[∘]:=

$$\begin{aligned}
&\mathrm{R}[\mathrm{QU}, kk_] := \mathrm{R}[\mathrm{QU}, kk] = \mathbb{E}\Big[-\frac{\hbar\, a_2\, t_1}{\gamma}, \hbar\, x_2\, y_1,\\
&\quad \mathrm{Series}\Big[\mathbb{e}^{\hbar\,\gamma^{-1}\,t_1\,a_2-\hbar\,y_1\,x_2}\left(\mathbb{e}^{\hbar\,b_1\,a_2}\,\mathbb{e}_{q_\hbar,kk}[\hbar\,y_1\,x_2]\ /.\ b_1\to\gamma^{-1}\left(\epsilon\,a_1-t_1\right)\right),\ \{\epsilon, 0, kk\}\Big]\Big];\\
&\mathrm{tR}_{i_,j_} := \mathrm{R}[\$U, \$k]\ /.\ \left\{\left(v:t\,|\,T\,|\,y\,|\,a\,|\,x\right)_1\to v_i,\ \left(v:t\,|\,T\,|\,y\,|\,a\,|\,x\right)_2\to v_j\right\};\\
&\overline{\mathrm{tR}}_{i_,j_} := \overline{\mathrm{tR}}_{i,j} = \mathrm{tR}_{i,j}\sim B_j\sim \mathrm{tS}_j;
\end{aligned}$$

SL2Portfolio

In[∘]:= $\left\{\mathrm{tR}_{1,2},\ \overline{\mathrm{tR}}_{1,2}\right\}$

SL2Portfolio

Out[∘]=

$$\begin{aligned}
&\Big\{\mathbb{E}\Big[-\frac{\hbar\,a_2\,t_1}{\gamma}, \hbar\,x_2\,y_1, 1+\left(\frac{\hbar\,a_1\,a_2}{\gamma}-\frac{1}{4}\,\gamma\,\hbar^3\,x_2^2\,y_1^2\right)\epsilon+O[\epsilon]^2\Big], \mathbb{E}\Big[\frac{\hbar\,a_2\,t_1}{\gamma}, -\frac{\hbar\,x_2\,y_1}{T_1},\\
&\quad 1+\frac{1}{4\,\gamma\,T_1^2}\left(-4\,\hbar\,a_1\,a_2\,T_1^2-4\,\gamma\,\hbar^2\,a_1\,T_1\,x_2\,y_1-4\,\gamma\,\hbar^2\,a_2\,T_1\,x_2\,y_1-3\,\gamma^2\,\hbar^3\,x_2^2\,y_1^2\right)\epsilon+O[\epsilon]^2\Big]\Big\}
\end{aligned}$$

SL2Portfolio

tC is the counterclockwise spinner; $\overline{\mathrm{tC}}$ is its inverse.

SL2Portfolio

In[∘]:=

$$\begin{aligned}
&\mathrm{tC}_{i_} := \mathbb{E}\left[0, 0, T_i^{1/2}\,\mathbb{e}^{-\epsilon\,a_i\,\hbar}+0_{\$k}\right];\\
&\overline{\mathrm{tC}}_{i_} := \mathbb{E}\left[0, 0, T_i^{-1/2}\,\mathbb{e}^{\epsilon\,a_i\,\hbar}+0_{\$k}\right];
\end{aligned}$$

SL2Portfolio

In[∘]:= $\mathrm{Block}\left[\{\$k = 3\}, \left\{\mathrm{tC}_1, \overline{\mathrm{tC}}_2\right\}\right]$

SL2Portfolio

Out[∘]=

$$\begin{aligned}
&\Big\{\mathbb{E}\Big[0, 0, \sqrt{T_1}-\hbar\,a_1\sqrt{T_1}\,\epsilon+\frac{1}{2}\,\hbar^2\,a_1^2\sqrt{T_1}\,\epsilon^2-\frac{1}{6}\left(\hbar^3\,a_1^3\sqrt{T_1}\right)\epsilon^3+O[\epsilon]^4\Big],\\
&\quad \mathbb{E}\Big[0, 0, \frac{1}{\sqrt{T_2}}+\frac{\hbar\,a_2\,\epsilon}{\sqrt{T_2}}+\frac{\hbar^2\,a_2^2\,\epsilon^2}{2\sqrt{T_2}}+\frac{\hbar^3\,a_2^3\,\epsilon^3}{6\sqrt{T_2}}+O[\epsilon]^4\Big]\Big\}
\end{aligned}$$

SL2Portfolio

In[∘]:=

$$\begin{aligned}
&\mathrm{Kink}[\mathrm{QU}, kk_] := \mathrm{Kink}[\mathrm{QU}, kk] = \mathrm{Block}\left[\{\$k = kk\}, \left(\mathrm{tR}_{1,3}\,\overline{\mathrm{tC}}_2\right)\sim B_{1,2}\sim \mathrm{tm}_{1,2\to 1}\sim B_{1,3}\sim \mathrm{tm}_{1,3\to 1}\right];\\
&\mathrm{tKink}_{i_} := \mathrm{Kink}[\$U, \$k]\ /.\ \left\{\left(v:t\,|\,T\,|\,y\,|\,a\,|\,x\right)_1\to v_i\right\};\\
&\overline{\mathrm{Kink}}[\mathrm{QU}, kk_] := \overline{\mathrm{Kink}}[\mathrm{QU}, kk] = \mathrm{Block}\left[\{\$k = kk\}, \left(\overline{\mathrm{tR}}_{1,3}\,\mathrm{tC}_2\right)\sim B_{1,2}\sim \mathrm{tm}_{1,2\to 1}\sim B_{1,3}\sim \mathrm{tm}_{1,3\to 1}\right];\\
&\overline{\mathrm{tKink}}_{i_} := \overline{\mathrm{Kink}}[\$U, \$k]\ /.\ \left\{\left(v:t\,|\,T\,|\,y\,|\,a\,|\,x\right)_1\to v_i\right\}
\end{aligned}$$

SL2Portfolio

Alternative Algorithms

SL2Portfolio

In[∘]:=

$$\begin{aligned}
&\lambda_{\mathrm{alt},k_}[\mathrm{CU}] := \mathrm{If}\Big[k == 0, 1, \mathrm{Module}\Big[\{eq, d, b, c, so\},\\
&\quad eq = \rho@\mathbb{e}^{\xi\,x_{\mathrm{CU}}}.\rho@\mathbb{e}^{\eta\,y_{\mathrm{CU}}} == \rho@\mathbb{e}^{d\,y_{\mathrm{CU}}}.\rho@\mathbb{e}^{c\,(t\,1_{\mathrm{CU}}-2\,\epsilon\,a_{\mathrm{CU}})}.\rho@\mathbb{e}^{b\,x_{\mathrm{CU}}};\\
&\quad \{so\} = \mathrm{Solve}[\mathrm{Thread}[\mathrm{Flatten}\ /@\ eq], \{d, b, c\}]\ /.\ C@1\to 0;\\
&\quad \mathrm{Series}\Big[\mathbb{e}^{-\eta\,y-\xi\,x+\eta\,\xi\,t+c\,t+d\,y-2\,\epsilon\,c\,a+b\,x}\ /.\ so,\ \{\epsilon, 0, k\}\Big]\Big]\Big];
\end{aligned}$$

SL2Portfolio

The Trefoil

SL2Portfolio

In[∘]:=
```
Block[{$k = 1},
 Z = tR_{1,5} tR_{6,2} tR_{3,7} tC̄_4 tKink̄_8 tKink̄_9 tKink̄_10;
 Do[Z = Z~B_{1,k}~tm_{1,k→1}, {k, 2, 10}]; Z]
```

SL2Portfolio

Out[∘]=

$$\mathbb{E}\left[0,\,0,\,\frac{T_1}{1-T_1+T_1^2}+\left(\left(-2\,\hbar\,a_1\,T_1-\gamma\,\hbar\,T_1^2+2\,\hbar\,a_1\,T_1^2+2\,\gamma\,\hbar\,T_1^3-3\,\gamma\,\hbar\,T_1^4-2\,\hbar\,a_1\,T_1^4+2\,\gamma\,\hbar\,T_1^5+2\,\hbar\,a_1\,T_1^5-2\,\gamma\,\hbar^2\,T_1\,x_1\,y_1-2\,\gamma\,\hbar^2\,T_1^4\,x_1\,y_1\right)\epsilon\right)\Big/\left(1-3\,T_1+6\,T_1^2-7\,T_1^3+6\,T_1^4-3\,T_1^5+T_1^6\right)+O[\epsilon]^2\right]$$